ADM and BSSN variables for the Kasner metric

A standard form of the Kasner metric is given by

$$ds^2 = -dt^2 + t^{2p_1} dx^2 + t^{2p_2} dy^2 + t^{2p_3} dz^2$$

where p_1, p_2 and p_3 are constants subject to

$$\begin{aligned} 1 &= p_1 + p_2 + p_3 \\ 1 &= p_1^1 + p_2^2 + p_2^3 \end{aligned}$$

The following Cadabra codes compute various quantities defined in the ADM and BSSN formulations of the Einstein equations.

All of the results are exactly as expected (what else could it give?).

None of this is new – the main point of this whole exercise is to use a familar metric to explore how standard computations can be implemented using Cadabra.

None of these results are used by the main evolution codes (in the directories `adm` and `bssn`) other than to set the initial data (at $t = 1$).

The code that sets the initial data was written by hand (as opposed to the Cadabra codes that generates the Ada procedures used in the evolution codes).

```
{t,x,y,z}::Coordinate.
{a,b,c,d,e,f,i,j,k,l,m,n,o,p,q,r,s,u#}::Indices(position=independent,values={t,x,y,z}).

\partial{#}::PartialDerivative;

{p1,p2,p3}::Symbol.

p1::LaTeXForm("p_1").
p2::LaTeXForm("p_2").
p3::LaTeXForm("p_3").

gBar{#}::LaTeXForm("{\bar g}").
ABar{#}::LaTeXForm("{\bar A}").
Aab{#}::LaTeXForm("{A}").
phi::LaTeXForm("{\phi}").

g_{a b}::Metric.
g^{a b}::InverseMetric.

g_{a b}::Depends(\partial{#}).

# -----------------------------------------------------------------
# rules used when evaluating components

DtRule := {D^{t} -> 1}.      # components of d/dt, zero shift & unit lapse

gabRule := { g_{t t} = gtt,
             g_{x x} = gxx,
             g_{y y} = gyy,
             g_{z z} = gzz }.

# -----------------------------------------------------------------
# the Kasner metric

gab := { gtt -> -1,
         gxx -> t**(2*p1),
         gyy -> t**(2*p2),
         gzz -> t**(2*p3),
```

```
        gxy -> 0,
        gxz -> 0,
        gyz -> 0,
        gtx -> 0,
        gty -> 0,
        gtz -> 0}.

# ----------------------------------------------------------------
# standard definitions

Detg := g ->  gxx gyy gzz - gxx gyz gyz
            - gxy gxy gzz + gxy gxz gyz
            + gxz gxy gyz - gxz gxz gyy.

Gamma := \Gamma^{a}_{b c} ->
         (1/2) g^{a e} (   \partial_{b}{g_{e c}}
                         + \partial_{c}{g_{b e}}
                         - \partial_{e}{g_{b c}}).

Rabcd := R^{a}_{b c d} ->
           \partial_{c}{\Gamma^{a}_{b d}} + \Gamma^{a}_{e c} \Gamma^{e}_{b d}
         - \partial_{d}{\Gamma^{a}_{b c}} - \Gamma^{a}_{e d} \Gamma^{e}_{b c}.


Rab := R_{a b} -> R^{c}_{a c b}.

Kab := K_{a b} -> - (1/2) D^{c} \partial_{c}{g_{a b}} / N.

# ----------------------------------------------------------------
# the BSSN variables

trK   := K          -> g^{a b} K_{a b}.
Aab   := Aab_{a b}  -> K_{a b} - (1/3) g_{a b} K.
gBar  := gBar_{a b} -> g_{a b} / (g**(1/3)).
ABar  := ABar_{a b} -> (K_{a b} - (1/3) g_{a b} K) / (g**(1/3)).
phi   := phi        -> (1/12) \log(g).

# ----------------------------------------------------------------
```

```
# basic objects

substitute (gabRule, gab)
substitute (Detg,    gab)

complete   (gabRule, $g^{a b}$)                                     # cdb(gabRule,gabRule)

substitute (Rabcd,   Gamma)
substitute (Rab,     Rabcd)

# -----------------------------------------------------------------
# convert to BSSN

substitute (gBar,  Detg)                                            # cdb (gBar.01,gBar)

substitute (Aab,   trK)                                             # cdb (Aab.01,Aab)
substitute (Aab,   Kab)                                             # cdb (Aab.02,Aab)

substitute (ABar,  trK)                                             # cdb (ABar.01,ABar)
substitute (ABar,  Kab)                                             # cdb (ABar.02,ABar)
substitute (ABar,  Detg)                                            # cdb (ABar.03,ABar)

substitute (phi,   Detg)                                            # cdb (phi.01,phi)

# -----------------------------------------------------------------
# now evaluate the components

evaluate   (gab,   gabRule+DtRule, rhsonly=True)                    # cdb (gab,gab)
evaluate   (Gamma, gabRule+DtRule, rhsonly=True)                    # cdb (Gamma,Gamma)
evaluate   (Rabcd, gabRule+DtRule, rhsonly=True)                    # cdb (Rabcd,Rabcd)
evaluate   (Rab,   gabRule+DtRule, rhsonly=True)                    # cdb (Rab,Rab)
evaluate   (Kab,   gabRule+DtRule, rhsonly=True)                    # cdb (Kab,Kab)
evaluate   (trK,   gabRule+DtRule, rhsonly=True)                    # cdb (trK,trK)

evaluate   (gBar,  gabRule+DtRule, rhsonly=True)                    # cdb (gBar.02,gBar)
evaluate   (Aab,   gabRule+DtRule, rhsonly=True)                    # cdb (Aab.03,Aab)
evaluate   (ABar,  gabRule+DtRule, rhsonly=True)                    # cdb (ABar.04,ABar)
```

```
evaluate    (phi,    gabRule+DtRule, rhsonly=True)                          # cdb (phi.02,phi)
```

$$\left[gtt \to -1,\ gxx \to t^{2p_1},\ gyy \to t^{2p_2},\ gzz \to t^{2p_3},\ gxy \to 0,\ gxz \to 0,\ gyz \to 0,\ gtx \to 0,\ gty \to 0,\ gtz \to 0\right] \tag{gab}$$

$$\Gamma^{a}{}_{bc} \to \Box_{cb}{}^{a} \begin{cases} \Box_{zt}{}^{z} = p_3 t^{-1} \\ \Box_{yt}{}^{y} = p_2 t^{-1} \\ \Box_{xt}{}^{x} = p_1 t^{-1} \\ \Box_{tz}{}^{z} = p_3 t^{-1} \\ \Box_{ty}{}^{y} = p_2 t^{-1} \\ \Box_{tx}{}^{x} = p_1 t^{-1} \\ \Box_{zz}{}^{t} = p_3 t^{(2p_3-1)} \\ \Box_{yy}{}^{t} = p_2 t^{(2p_2-1)} \\ \Box_{xx}{}^{t} = p_1 t^{(2p_1-1)} \end{cases} \tag{Gamma}$$

$$
R^{a}{}_{bcd} \to \Box_{db}{}^{a}{}_{c}
\begin{cases}
\Box_{xx}{}^{t}{}_{t} = p_1 t^{(2p_1-2)}\left(p_1 - 1\right) \\
\Box_{yy}{}^{t}{}_{t} = p_2 t^{(2p_2-2)}\left(p_2 - 1\right) \\
\Box_{zz}{}^{t}{}_{t} = p_3 t^{(2p_3-2)}\left(p_3 - 1\right) \\
\Box_{xt}{}^{x}{}_{t} = p_1\left(p_1 - 1\right) t^{-2} \\
\Box_{yt}{}^{y}{}_{t} = p_2\left(p_2 - 1\right) t^{-2} \\
\Box_{zt}{}^{z}{}_{t} = p_3\left(p_3 - 1\right) t^{-2} \\
\Box_{tz}{}^{t}{}_{z} = -p_3 t^{(2p_3-2)}\left(p_3 - 1\right) \\
\Box_{ty}{}^{t}{}_{y} = -p_2 t^{(2p_2-2)}\left(p_2 - 1\right) \\
\Box_{tx}{}^{t}{}_{x} = -p_1 t^{(2p_1-2)}\left(p_1 - 1\right) \\
\Box_{zz}{}^{y}{}_{y} = p_2 p_3 t^{(2p_3-2)} \\
\Box_{zz}{}^{x}{}_{x} = p_1 p_3 t^{(2p_3-2)} \\
\Box_{yy}{}^{z}{}_{z} = p_2 p_3 t^{(2p_2-2)} \\
\Box_{yy}{}^{x}{}_{x} = p_1 p_2 t^{(2p_2-2)} \\
\Box_{xx}{}^{z}{}_{z} = p_1 p_3 t^{(2p_1-2)} \\
\Box_{xx}{}^{y}{}_{y} = p_1 p_2 t^{(2p_1-2)} \\
\Box_{tt}{}^{x}{}_{x} = p_1\left(-p_1 + 1\right) t^{-2} \\
\Box_{tt}{}^{y}{}_{y} = p_2\left(-p_2 + 1\right) t^{-2} \\
\Box_{tt}{}^{z}{}_{z} = p_3\left(-p_3 + 1\right) t^{-2} \\
\Box_{yz}{}^{y}{}_{z} = -p_2 p_3 t^{(2p_3-2)} \\
\Box_{xz}{}^{x}{}_{z} = -p_1 p_3 t^{(2p_3-2)} \\
\Box_{zy}{}^{z}{}_{y} = -p_2 p_3 t^{(2p_2-2)} \\
\Box_{xy}{}^{x}{}_{y} = -p_1 p_2 t^{(2p_2-2)} \\
\Box_{zx}{}^{z}{}_{x} = -p_1 p_3 t^{(2p_1-2)} \\
\Box_{yx}{}^{y}{}_{x} = -p_1 p_2 t^{(2p_1-2)}
\end{cases}
\tag{Rabcd}
$$

$$
R_{ab} \to \Box_{ba}
\begin{cases}
\Box_{xx} = p_1 t^{(2p_1-2)}\left(p_1 + p_2 + p_3 - 1\right) \\
\Box_{yy} = p_2 t^{(2p_2-2)}\left(p_1 + p_2 + p_3 - 1\right) \\
\Box_{zz} = p_3 t^{(2p_3-2)}\left(p_1 + p_2 + p_3 - 1\right) \\
\Box_{tt} = \left(-p_1{}^2 + p_1 - p_2{}^2 + p_2 - p_3{}^2 + p_3\right) t^{-2}
\end{cases}
\tag{Rab}
$$

$$
K_{ab} \to \Box_{ab}
\begin{cases}
\Box_{zz} = -p_3 t^{(2p_3-1)} N^{-1} \\
\Box_{yy} = -p_2 t^{(2p_2-1)} N^{-1} \\
\Box_{xx} = -p_1 t^{(2p_1-1)} N^{-1}
\end{cases}
\tag{Kab}
$$

$$K \to -K_{tt} + t^{-2p_3}K_{zz} + t^{-2p_2}K_{yy} + t^{-2p_1}K_{xx} \tag{trK}$$

$$\bar{g}_{ab} \to g_{ab}\big(t^{2p_1}t^{2p_2}t^{2p_3}\big)^{-\frac{1}{3}} \tag{gBar.01}$$

$$\bar{g}_{ab} \to \Box_{ab}\begin{cases} \Box_{tt} = -t^{(2p_1+2p_2+2p_3)^{-\frac{1}{3}}} \\ \Box_{xx} = t^{2p_1}t^{(2p_1+2p_2+2p_3)^{-\frac{1}{3}}} \\ \Box_{yy} = t^{2p_2}t^{(2p_1+2p_2+2p_3)^{-\frac{1}{3}}} \\ \Box_{zz} = t^{2p_3}t^{(2p_1+2p_2+2p_3)^{-\frac{1}{3}}} \end{cases} \tag{gBar.02}$$

$$\phi \to \frac{1}{12}\log\left(t^{2p_1}t^{2p_2}t^{2p_3}\right) \tag{phi.01}$$

$$\phi \to \frac{1}{12}\log\left(t^{(2p_1+2p_2+2p_3)}\right) \tag{phi.02}$$

$$A_{ab} \to K_{ab} - \frac{1}{3}g_{ab}g^{cd}K_{cd} \tag{Aab.01}$$

$$A_{ab} \to -\frac{1}{2}D^c\partial_c g_{ab}N^{-1} + \frac{1}{6}g_{ab}g^{cd}D^e\partial_e g_{cd}N^{-1} \tag{Aab.02}$$

$$A_{ab} \to \Box_{ab}\begin{cases} \Box_{zz} = \frac{1}{3}t^{(2p_3-1)}\left(p_1+p_2-2p_3\right)N^{-1} \\ \Box_{yy} = \frac{1}{3}t^{(2p_2-1)}\left(p_1-2p_2+p_3\right)N^{-1} \\ \Box_{xx} = \frac{1}{3}t^{(2p_1-1)}\left(-2p_1+p_2+p_3\right)N^{-1} \\ \Box_{tt} = -\frac{1}{3}\left(p_1+p_2+p_3\right)(Nt)^{-1} \end{cases} \tag{Aab.03}$$

$$\bar{A}_{ab} \to \left(K_{ab} - \frac{1}{3} g_{ab} g^{cd} K_{cd}\right) g^{-\frac{1}{3}} \tag{ABar.01}$$

$$\bar{A}_{ab} \to \left(-\frac{1}{2} D^c \partial_c g_{ab} N^{-1} + \frac{1}{6} g_{ab} g^{cd} D^e \partial_e g_{cd} N^{-1}\right) g^{-\frac{1}{3}} \tag{ABar.02}$$

$$\bar{A}_{ab} \to \left(-\frac{1}{2} D^c \partial_c g_{ab} N^{-1} + \frac{1}{6} g_{ab} g^{cd} D^e \partial_e g_{cd} N^{-1}\right) \left(t^{2p_1} t^{2p_2} t^{2p_3}\right)^{-\frac{1}{3}} \tag{ABar.03}$$

$$\bar{A}_{ab} \to \square_{ab} \begin{cases} \square_{zz} = \frac{1}{3} t^{(2p_3-1)} \left(p_1 + p_2 - 2p_3\right) \left(N t^{(2p_1+2p_2+2p_3)^{\frac{1}{3}}}\right)^{-1} \\ \square_{yy} = \frac{1}{3} t^{(2p_2-1)} \left(p_1 - 2p_2 + p_3\right) \left(N t^{(2p_1+2p_2+2p_3)^{\frac{1}{3}}}\right)^{-1} \\ \square_{xx} = \frac{1}{3} t^{(2p_1-1)} \left(-2p_1 + p_2 + p_3\right) \left(N t^{(2p_1+2p_2+2p_3)^{\frac{1}{3}}}\right)^{-1} \\ \square_{tt} = -\frac{1}{3} \left(p_1 + p_2 + p_3\right) \left(N t t^{(2p_1+2p_2+2p_3)^{\frac{1}{3}}}\right)^{-1} \end{cases} \tag{ABar.04}$$